<?xml version="1.0" encoding="UTF-8"?>

<!DOCTYPE xsl:stylesheet SYSTEM "lang://pages">

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">	
	
	<xsl:include href="../admin.xsl" />
	<xsl:include href="../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;pages')/data/response" mode="cp-path" />
		</div>
		
		<h1>Статичные страницы</h1>
		
		<p>
			<a class="add" href="{$baseURI}/admin/pages/add">Создать новую</a>
		</p>
		
		<xsl:apply-templates select="message" />
		
		<table class="standart" cellspacing="0" cellpadding="0">
			<tbody>
				<xsl:apply-templates select="object[@type='page']" />
			</tbody>
		</table>
	</xsl:template>
	
	<xsl:template match="object[@type='page']">
		<xsl:param name="alias" select="property[@name='alias']/value" />
		
		<xsl:if test="position() = 1">
			<tr>
				<th>Заголовок</th>
				<th>Опции</th>
			</tr>
		</xsl:if>
		
		<tr>
			<td>
				<a href="{$baseURI}/page/{$alias}" class="mainlink" title="Просмотреть">
					<xsl:value-of select="property[@name='title']/value" />
				</a>
			</td>
			<td>
				<a href="{$baseURI}/admin/pages/edit/{@id}">Редактировать</a>
				<xsl:text> ∙ </xsl:text>
				<a href="{$baseURI}/admin/pages/delete/{@id}?scode={$scode}" onclick="if(!confirm('Вы действительно хотите удалить эту страницу?'))return false;">Удалить</a>
			</td>
		</tr>
	</xsl:template>
	
</xsl:stylesheet>